An Integral Involving the Product of Five Bessel Functions

Solution of Problem 04-003 by Michael Renardy (Virginia Polytech Institute and State University).

We take advantage of the asymptotic expansion
\[ J_0(x) = \sqrt{\frac{2}{\pi x}} \left\{ \cos(x - \pi/4) \left( 1 - \frac{9}{128x^2} + O(x^{-4}) \right) + \sin(x - \pi/4) \left( \frac{1}{8x} + O(x^{-3}) \right) \right\} \]
and the fact that the resulting approximation to the integrand can be integrated in closed form in terms of Fresnel integrals. The result is
\[ \int_0^\infty x J_0(x\sqrt{2}) J_0(x\sqrt{3}) J_0(x\sqrt{5}) J_0(x\sqrt{7}) J_0(x\sqrt{11}) \, dx \approx 0.061064349909, \]
and a listing of the Mathematica code is given below. Also, a Mathematica notebook is available as a download option. Evidence of convergence is provided by the fact that when the transition point is changed from 100 to 200, the value of \( p_1 + p_2 \) is unchanged in the number of digits shown.

\begin{verbatim}
ja[x_] = Sqrt[2/Pi/x](Cos[x-Pi/4](1-9/128/x^2)+Sin[x-Pi/4]/8/x)
f[x_] = x*BesselJ[0,x*Sqrt[2]]*BesselJ[0,x*Sqrt[3]]*BesselJ[0,x*Sqrt[5]]*BesselJ[0,x*Sqrt[7]]*BesselJ[0,x*Sqrt[11]]
fa[x_] = x*ja[x*Sqrt[2]]*ja[x*Sqrt[3]]*ja[x*Sqrt[5]]*ja[x*Sqrt[7]]*ja[x*Sqrt[11]]
fa[x_] = TrigReduce[fa[x]];
fa[x_] = N[fa[x],20];
fa[x_] = TrigExpand[fa[x]];
fb1[x_] = Coefficient[PowerExpand[fa[x^2]], 1/x^3]/x^3 + Coefficient[PowerExpand[fa[x^2]], 1/x^5]/x^5 + Coefficient[PowerExpand[fa[x^2]], 1/x^7]/x^7;
fb2[x_] = PowerExpand[fb1[Sqrt[x]]];
p1 = Sum[NIntegrate[f[x], {x,n-1,n}, WorkingPrecision -> 20], {n,1,100}]
p2 = N[Integrate[fb2[x], {x,100,Infinity}], 20]
p1+p2
\end{verbatim}